

Structural Equation Modeling

M&S Lecture 2, 2017

Measurement Invariance and Path Analysis

Ellen Hamaker

Methodology and Statistics

Faculty of Social and Behavioural Sciences

Outline

- Multiple group factor analysis
- Path analysis

Multiple group factor analysis

Thus far we analyzed the **covariance structure**.

We can also analyze the **mean structure**.

Means may be of interest when:

- a) we have **multiple groups**
- b) we have **multiple occasions** (longitudinal data)



REMEMBER: a latent variable does not have an intrinsic **scale**. We assign it a scale by **scaling** the factor (either through fixing a factor loading, or through fixing its variance).

The **latent mean** is also unknown (unidentified).

However, we can estimate the **difference in latent means**, either between groups, or between occasions.

E.g., intelligence research


An important topic in **multiple group CFA** is which of the groups scores higher on the latent variable: This is very important in for instance **intelligence research**.



Related to this topic is the question whether a test allows us to make **fair comparisons between individuals from different groups**.

Is a test fair?


In a factor model, the observed variable y is regressed on the latent variable η . Hence, based on some one's standing on the underlying dimension, (s)he has a particular expected score on the observed variable.



Biased test

Biased implies that two individuals (from **different groups**) with the **same latent score** have **different (expected) observed scores**.


This can be due to a different **intercept**...



Biased test

... or to a different **slope** (= factor loading):

$$y_i = v_g + \lambda_g \eta_i + \varepsilon_i$$



Covariance and mean structures & likelihood

Measurement equations:

$$\text{Group 1: } y_i = v_1 + \Lambda_1 \eta_i + \varepsilon_i$$

$$\text{Group 2: } y_i = v_2 + \Lambda_2 \eta_i + \varepsilon_i$$

Covariance structures:

$$\text{Group 1: } \Sigma_1 = \Lambda_1 \Psi_1 \Lambda_1^T + \Theta_1$$

$$\text{Group 2: } \Sigma_2 = \Lambda_2 \Psi_2 \Lambda_2^T + \Theta_2$$

Mean structures:

$$\text{Group 1: } \mu_1 = v_1 + \Lambda_1 \alpha_1$$

$$\text{Group 2: } \mu_2 = v_2 + \Lambda_2 \alpha_2$$

$$\text{Log likelihood: } \log L = \sum_{g=1}^G \log L_g$$

where:

$$\log L_g = c - \frac{N_g}{2} \log |\Sigma_g| - \frac{N_g}{2} \text{tr}(\mathbf{S}_g \Sigma_g^{-1}) - \frac{N_g}{2} (\mathbf{m}_g - \mu_g)^T \Sigma_g^{-1} (\mathbf{m}_g - \mu_g)$$

Measurement invariance

To make sure that the test is **unbiased** (=fair), we have to test whether:


- 1) **factor loadings** are **identical** across the groups → **weak factorial invariance**: $\Lambda_1 = \Lambda_2$
- 2) the **intercepts** are **identical** across the groups → **strong factorial invariance**: $\nu_1 = \nu_2$

Strong factorial invariance implies that:

- Two people from different groups with the **same latent scores**, have the **same expected observed scores**
- **Group differences in observed means** are the result of **differences in latent means**
- The test allows for a fair comparison between groups, and between people from different groups.

Example: Wicherts & Dolan, 2010

A **4-factor model** for subtests in the RAKIT (an intelligence test):



Sample statistics, parameters, and df

Per group there are $12 \times 13 / 2 = 78$ unique elements in the covariance matrix, and 12 means, so there are 90 sample statistics.

Per group we estimate:


- 12 means
- 12 residual variances
- 4 factor variances
- 6 factor covariances
- 8 factor loadings

So 42 parameters in total.

Hence there are $90 - 42 = 48$ df per group.

Majority group

Specifying this model in Mplus for the **majority group only**:

A screenshot of a text editor window titled "Majority.inp". The window contains Mplus input code. The code is organized into sections: TITLE, DATA, VARIABLE, MODEL, and OUTPUT. The VARIABLE section includes a list of variable names and a USEVARIABLES statement. The MODEL section contains four factor loadings. The OUTPUT section specifies the output to be generated.

```
TITLE: Majority

DATA:  TYPE IS MEANS COVARIANCE;
       FILE IS majority.dat;
       NOBSEVATIONS ARE 196;

VARIABLE: NAMES = CLO EXC MSP VME MAZ ANA QUA DIS LEN HFI IDP STT;
          USEVARIABLES ARE ANA VME QUA MSP LEN DIS EXC MAZ HFI CLO IDP STT;

MODEL:  Hybrid BY ANA VME QUA;
        Memory BY MSP LEN;
        Visual BY DIS EXC MAZ HFI CLO;
        Retrieval BY IDP STT;

OUTPUT: SAMPSTAT TECH1 TECH4 MOD(4) STANDARDIZED;
```

NOTE: The statement USEVARIABLES can be used to make a selection of the variables from the original file, or to change the order of the variables.

Oops?

If we run this factor model, Mplus gives the following **warning**:

THE MODEL ESTIMATION TERMINATED NORMALLY

WARNING: THE LATENT VARIABLE COVARIANCE MATRIX (PSI) IS NOT POSITIVE DEFINITE. THIS COULD INDICATE A NEGATIVE VARIANCE/RESIDUAL VARIANCE FOR LATENT VARIABLE, **A CORRELATION GREATER OR EQUAL TO ONE BETWEEN TWO LATENT VARIABLES**, OR A LINEAR DEPENDENCY AMONG MORE THAN TWO LATENT VARIABLES. CHECK THE TECH4 OUTPUT FOR MORE INFORMATION.
PROBLEM INVOLVING VARIABLE VISUAL.

In the **TECH4 output**, we get the correlations matrix of the latent variables:

ESTIMATED CORRELATION MATRIX FOR THE LATENT VARIABLES

Correlation
of almost 1

	HYBRID	MEMORY	VISUAL	RETRIEVA
HYBRID	1.000			
MEMORY	0.876	1.000		
VISUAL	0.966	0.677	1.000	
RETRIEVA	0.437	0.418	0.426	1.000

Model fit majority

Chi-Square Test of Model Fit

Value	81.579
Degrees of Freedom	48
P-Value	0.0018

RMSEA (Root Mean Square Error Of Approximation)

Estimate	0.060	
90 Percent C.I.	0.036	0.082
Probability RMSEA <= .05	0.224	

CFI/TLI

CFI	0.908
TLI	0.874

Chi-Square Test of Model Fit for the Baseline Model

Value	431.961
Degrees of Freedom	66
P-Value	0.0000

SRMR (Standardized Root Mean Square Residual)

Value	0.051
-------	-------

Model fit minority

Chi-Square Test of Model Fit

Value	71.904
Degrees of Freedom	48
P-Value	0.0143

RMSEA (Root Mean Square Error Of Approximation)

Estimate	0.062	
90 Percent C.I.	0.028	0.090
Probability RMSEA <= .05	0.246	

CFI/TLI

CFI	0.921
TLI	0.892

Chi-Square Test of Model Fit for the Baseline Model

Value	370.034
Degrees of Freedom	66
P-Value	0.0000

SRMR (Standardized Root Mean Square Residual)

Value	0.057
-------	-------

Measurement invariance

Models to run & compare:

1. Free model: ***Configural invariance***
2. Constrain **factor loadings** (Λ) across groups: ***Weak factorial invariance***
3. Constrain **intercepts** (ν) across groups (while freeing latent means): ***Strong factorial invariance***

(In addition the residual variances can be constrained across the groups.

This is called ***strict factorial invariance***: $f(y|\eta, g) = f(y|\eta)$. Note that there can still be differences between the groups (in α and Ψ).)

Testing for weak factorial invariance

There is some discussion in the literature about whether this procedure is correct: Some claim that if (accidentally) you use a biased indicator for scaling, this procedure does not work...

However:

- If there are **no biased indicators** (e.g., strong factorial invariance holds), it does work
- If there are biased indicators, it does not matter whether you use **a biased or an unbiased** indicator for scaling: These models are **statistically indential** (i.e., they have the same fit, and thus lead to same chi-square difference tests)

Alternative 1: Just imposes the constraints, and look at the Modification Indices when the model does not fit (top-down procedure; cf. Byrne).

Alternative 2: Test each factor loading separately, using different factor loadings for scaling (not doable when there are many indicators...)

Model 1: Overruling the Mplus defaults

```
MajMin1.inp

TITLE: Majority and minority

DATA:  NGROUPS = 2;
      TYPE IS MEANS COVARIANCE;
      FILE IS majmin.dat;
      NOBSEVATIONS ARE 196 131;

VARIABLE: NAMES = CLO EXC MSP VME MAZ ANA QUA DIS LEN HFI IDP STT;
          USEVARIABLES ARE ANA VME QUA MSP LEN DIS EXC MAZ HFI CLO IDP STT;

MODEL:  Hybrid BY ANA VME QUA;
        Memory BY MSP LEN;
        Visual BY DIS EXC MAZ HFI CLO;
        Retrieval BY IDP STT;

MODEL G2: Hybrid BY ANA VME QUA;
          Hybrid BY ANA@1;
          Memory BY MSP LEN;
          Memory BY MSP@1;
          Visual BY DIS EXC MAZ HFI CLO;
          Visual BY DIS@1;
          Retrieval BY IDP STT;
          Retrieval BY IDP@1;
          [Hybrid-Retrieval@0];
          [ANA - STT];

OUTPUT: SAMPSTAT TECH1 TECH4 MOD(4) STANDARDIZED;
```

This is needed to **overrule the defaults**: Mplus automatically imposes the constraints for **strong factorial invariance** (i.e., equal factor loadings and intercepts). Note we also need to **scale the factors** in the second group!



Model 1: Configural Invariance

Write down the expressions for the covariance matrices and mean vectors of both groups.

Model 1: Model fit of both groups

Chi-Square Test of Model Fit

Value	153.483
Degrees of Freedom	96
P-Value	0.0002

Chi-Square Contributions From Each Group

G1	81.579
G2	71.904

NOTE: these values are the same as from the separate analyses (so there are absolutely no constraints across the two groups); they sum to the chi-square of this model

RMSEA (Root Mean Square Error Of Approximation)

Estimate	0.061
90 Percent C.I.	0.042 0.078
Probability RMSEA <= .05	0.163

CFI/TLI

CFI	0.914
TLI	0.882

...


SRMR (Standardized Root Mean Square Residual)

Value	0.053
-------	-------

BTW: Another very useful way of checking your model and seeing where the constraints and defaults are, is by looking at the **TECH1 output!**

Model 2: Equal factor loadings

In each group, 8 factor loadings were estimated. Hence, constraining these across groups gives us 8 df, such that model 2 should have $96 + 8 = 104$ df.



```
TITLE: Majority and minority

DATA:  NGROUPS = 2;
       TYPE IS MEANS COVARIANCE;
       FILE IS majmin.dat;
       NOOBSERVATIONS ARE 196 131;

VARIABLE: NAMES = CLO EXC MSP VME MAZ ANA QUA DIS LEN HFI IDP STT;
          USEVARIABLES ARE ANA VME QUA MSP LEN DIS EXC MAZ HFI CLO IDP STT;

MODEL:  Hybrid BY ANA VME QUA;
        Memory BY MSP LEN;
        Visual BY DIS EXC MAZ HFI CLO;
        Retrieval BY IDP STT;

MODEL G2: [Hybrid-Retrieval@0];
          [ANA - STT];

OUTPUT: SAMPSTAT TECH1 TECH4 MOD(4) STANDARDIZED;
```

Now we only overrule the defaults for the means: Fix latent means to zero (@0), and free the intercepts



Model 2: Weak Factorial Invariance

Write down the expressions for the covariance matrices and mean vectors of both groups.

Model 2: Equal factor loadings

Chi-Square Test of Model Fit

Value	158.124
Degrees of Freedom	104
P-Value	0.0005

Chi-square difference between this model and the previous model is:

$158 - 153 = 5$, with 8 df, which is **not significant**.


Hence, we do not have to reject the **null model** (Model 2 with equal factor loadings) in favor of the **alternative model** (Model 1 without constraints).

Put differently: We can **assume there are equal factor loadings**.

Model 3: Equal intercepts and factor loadings

In each group, 12 intercepts were estimated: if we would constrain these, we would be saying that the means across the two groups are equal.

However, what we want to say is that the **differences in observed means** are completely explained by **mean differences on the latent variables**.



Therefore, we **constrain the intercepts across the groups**, and **free the latent means in one of the groups**: these represent the **latent mean differences**.

Hence we have: $12 - 4 = 8$ parameters less in Model 3 compared to Model 2.



Model 3: Strong Factorial Invariance

Write down the expressions for the covariance matrices and mean vectors of both groups.



Nesting of model 2 and model 3

At first it may be confusing that in model 3 we **constrain the intercepts AND free latent means** in the second group: How does this lead to **nested models**?

It helps to write down the **modeled mean differences** between the two groups in each model.

In **model 2** we have: $\mu_2 - \mu_1 = \nu_2 - \nu_1$


This requires **12 additional parameters** (in addition to the 12 mean parameters in group 1).

In **model 3** we have: $\mu_2 - \mu_1 = \nu + \Lambda\alpha_2 - \nu = \Lambda\alpha_2$

This requires only **4 additional parameters** (in addition to the 12 mean parameters in group 1).

Model 3: Equal intercepts and factor loadings

This is the **default model that Mplus will fit** when doing a multiple group analysis: hence, the specification is easy:



```

MajMin3.inp
TITLE:  Majority and minority

DATA:  NGROUPS = 2;
       TYPE IS MEANS COVARIANCE;
       FILE IS majmin.dat;
       NOBSEVATIONS ARE 196 131;

VARIABLE:  NAMES = CLO EXC MSP VME MAZ ANA QUA DIS LEN HFI IDP STT;
           USEVARIABLES ARE ANA VME QUA MSP LEN DIS EXC MAZ HFI CLO IDP STT;

MODEL:    Hybrid BY ANA VME QUA;
           Memory BY MSP LEN;
           Visual BY DIS EXC MAZ HFI CLO;
           Retrieval BY IDP STT;

OUTPUT:  SAMPSTAT TECH1 TECH4 MOD(4) STANDARDIZED;
```



Model 3: Equal intercepts and factor loadings

Chi-Square Test of Model Fit

Value	230.660
Degrees of Freedom	112
P-Value	0.0000

Chi-square difference between this model and the previous model is:

$231 - 158 = 73$, with 8 df, which is **significant**.

Hence, we **cannot** impose the constraint of equal intercepts.

There is no strong factorial invariance!

To detect **sources of bias**, we look at the **modification indices**: they indicate how much the model would improve if we free this parameter.

Model 3: Check MIs for intercepts

MODINDICES (ALL).

Minimum M.I. value for printing the modification index 20.000

	M.I.	E.P.C.	Std E.P.C.	StdYX
E.P.C.				
Group G1				
BY Statements				
VISUAL BY VME	21.475	-0.980	-2.917	-0.522
Means/Intercepts/Thresholds				
[VME]	42.546	0.471	0.471	0.084
[QUA]	24.607	-0.430	-0.430	-0.089

Group G2

Means/Intercepts/Thresholds

[VME]	42.551	-9.159	-9.159	-1.773
[QUA]	24.609	5.078	5.078	1.006

Model 3 adjusted for bias: Check MIs again

Including: MODEL G2: [VME];

Model fit:

Value	185.151
Degrees of Freedom	111
P-Value	0.0000

Compared to model 2: $185 - 158 = 27$ with 7 df, $p=.0003$

Modification indices now indicate largest change for:

[MSP]	17.769	-0.488	-0.488	-0.102
[LEN]	17.770	0.138	0.138	0.027

Note it is **no longer** for the intercept of QUA!

That is because **MIs change**, if you change the model.

Model 3 further adjusted for bias

Including: **MODEL G2: [LEN];**

Model fit:

Value	166.477
Degrees of Freedom	110
P-Value	0.0004

Compared to model 2: $166 - 158 = 8$ with 6 df, $p=.238$

Now we have a model which has:

- equal factor loadings
- equal intercepts except for VME and LEN (to account for bias)

This allows us to **compare the latent means.**

Model 3 further adjusted for bias

Remember: we fixed the latent means in group 1 (majority) to zero, and estimated the latent means in group 2 (minority), which thus represent the **difference in latent means**.

Means

HYBRID	-4.184	0.472	-8.866	0.000
MEMORY	-0.626	0.635	-0.986	0.324
VISUAL	-4.155	0.464	-8.962	0.000
RETRIEVAL	-4.414	0.556	-7.945	0.000

Conclusion: Minority group scores on average significantly lower on Hybrid, Visual and Retrieval.

In addition, we should check the **intercepts that were not constrained**, to determine the direction of bias.

Nesting

It is not easy to see that Model 3 is **nested under** Model 2:

- a) we are **not only constraining** parameters (intercepts) to be equal
- b) **but also freeing** parameters (latent means)

This is called the **Reference-Group method**.

There are two alternative methods (which are **statistically equivalent!**):
the **Marker-variable method** and the **Effects-coding method**.

Marker-variable method

Model 1: Configural invariance

- fix **one of the intercepts** per factor to **zero** in each group (for mean structure); estimate all other intercepts and latent means freely in each group
- fix **one of the factor loadings** per factor to **one** in each group (for covariance structure); estimate all other factor loadings freely in each group

Model 2: Weak factorial invariance

- constrain factor loadings across the groups

Model 3: Strong factorial invariance

- constrain intercepts across the groups

Advantage: nesting is more obvious

Disadvantage: requires to overrule many defaults in Mplus

Effects-coding method

Model 1: Configural invariance

- intercepts are constrained to sum to zero per group; estimate latent means freely per group
- average factor loading is 1

Model 2: Weak factorial invariance

- constrain factor loadings across the groups

Model 3: Strong factorial invariance

- constrain intercepts across the groups

Advantage: latent and observed variables have a comparable scale

Disadvantage: makes no sense when the observed variables have very different scales, and it requires overruling defaults in Mplus

Outline

- Multiple group factor analysis
- **Path analysis**

And now for something completely different

Thus far, we have focused on **factor analysis**; another form of SEM is **path analysis**.

Path analysis is closely related to **regression analysis**: observed variables are used to predict other observed variable.

In path analysis, we may have:

- **more than one outcome variable**
- a variable that is both regressed on predictors, and a predictor of other variables (i.e., **be a mediator**).

It is also possible to do **path analysis with latent variables**.

Effect of corporal punishment

Corporal punishment is: the deliberate infliction of pain as retribution for an offence, or for the purpose of disciplining or reforming a wrongdoer or to change an undesirable attitude or behavior.



We are interested in how **corporal punishment** influences children's **psychological maladjustment**.

Sample: 175 children between the ages of 8 and 18.



Rohner, R. P., Bourque, S. L., & Elordi, C. A. (1996). Children's perceptions of corporal punishment, caretaker acceptance, and psychological adjustment in a poor, biracial southern community. *Journal of Marriage and the Family*, 58, 842-852.

Laws on corporal punishment



- Corporal punishment prohibited in **schools** and the **home**
- Corporal punishment prohibited in **schools** only
- Corporal punishment **not** prohibited



Research question: mediation

Specifically, is the effect of perceived harshness and perceived justness of physical punishment on children's psychological maladjustment **mediated** by perceived caretaker's rejection?

Variables:

Perceived harshness:	0 = never punished physically in any way 16 = punished more than 12 times a week, very hard
Perceived justness:	2 = very unfair and almost never deserved 8 = very fair and almost always deserved
Perceived rejection:	my mother does not really love me my mother ignores me as long as I do not bother her my mother goes out of her way to hurt my feelings
Psych. maladjustment:	higher score implies more problems

Mediation model: path diagram


sample statistics:

$$4 \times 5 / 2 = 10$$

free parameters:

1 covariance
2 variances
3 regression parameters
2 residual variances

$$df = 10 - 8 = 2$$





DIY

How can you get the **measurement equation** to be:

$$\begin{aligned}\mathbf{y}_i &= \nu + \Lambda \eta_i + \varepsilon_i \\ &= \eta_i\end{aligned}$$

DIY

Write down the **structural equation** for this model:

$$\eta_i = \alpha + B\eta_i + \zeta_i$$

DIY

The current expression for η_i is **recursive**. Rewrite it, such that η_i only appears on the left-hand side.

$$\eta_i = \alpha + B\eta_i + \zeta_i$$

Full structural equation model:


Measurement equation: $\mathbf{y}_i = \boldsymbol{\nu} + \boldsymbol{\Lambda} \boldsymbol{\eta}_i + \boldsymbol{\varepsilon}_i$

Structural equation: $\boldsymbol{\eta}_i = \boldsymbol{\alpha} + \boldsymbol{B} \boldsymbol{\eta}_i + \boldsymbol{\zeta}_i$

Mean structure: $\boldsymbol{\mu} = \boldsymbol{\nu} + \boldsymbol{\Lambda}(\boldsymbol{I} - \boldsymbol{B})^{-1} \boldsymbol{\alpha}$

Covariance structure: $\boldsymbol{\Sigma} = \boldsymbol{\Lambda}(\boldsymbol{I} - \boldsymbol{B})^{-1} \boldsymbol{\Psi}(\boldsymbol{I} - \boldsymbol{B})^{-1T} \boldsymbol{\Lambda}^T + \boldsymbol{\Theta}$

Model specification



```
CorPun.inp
TITLE:      Path model for mediation

DATA:       FILE IS CorPun.dat;
            TYPE IS COVA;
            NOBSERVATIONS ARE 175;

VARIABLE:   NAMES ARE harsh just reject malad;

MODEL:      harsh WITH just;
            ! NOTE: If you do not specify this covariance,
            ! it is still being estimated, but not reported
            reject ON harsh just;
            malad ON reject;

MODEL INDIRECT:
            malad IND reject just;
            malad IND reject harsh;

OUTPUT:     Tech1 Standardized MODINDICES (0);
```

Model fit

Chi-Square Test of Model Fit

Value	1.546
Degrees of Freedom	2
P-Value	0.4616

RMSEA (Root Mean Square Error Of Approximation)

Estimate	0.000	
90 Percent C.I.	0.000	0.139
Probability RMSEA \leq .05	0.598	

CFI/TLI

CFI	1.000
TLI	1.012

SRMR (Standardized Root Mean Square Residual)

Value	0.023
-------	-------

Parameter estimates

		Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
REJECT	ON				
HARSH		2.607	0.584	4.462	0.000
JUST		-4.395	0.763	-5.758	0.000
MALAD	ON				
REJECT		0.387	0.051	7.658	0.000
HARSH	WITH				
JUST		-0.259	0.262	-0.987	0.324
Variances					
HARSH		4.514	0.483	9.354	0.000
JUST		2.645	0.283	9.354	0.000
Residual Variances					
REJECT		268.220	28.674	9.354	0.000
MALAD		158.956	16.993	9.354	0.000

NOTE: this parameter is only **reported** because we explicitly asked Mplus for it

Standardized results

Mplus gives 3 forms of standardized results when requested:

- **STDYX Standardization:** uses the variances of the continuous latent variables as well as the variances of the background (=exogenous) and outcome (endogenous) variables
- **STDY Standardization:** uses the variances of the continuous latent variables and the variances of the outcome (endogenous) variables; this should be used if one has a binary (=dichotomous) predictor
- **STD Standardization:** uses the variances of the continuous latent variables for standardization

Here we use STDYX; we could also specifically ask for this with:

OUTPUT: STDYX

Indirect (=mediated) effects

There is **no direct path** from harshness to maladjustment, but there is an **indirect path** (mediated through rejection). Idem for justness.

TOTAL, TOTAL INDIRECT, SPECIFIC INDIRECT, AND DIRECT EFFECTS

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Effects from JUST to MALAD				
Sum of indirect	-1.700	0.369	-4.602	0.000
Specific indirect				
MALAD				
REJECT				
JUST	-1.700	0.369	-4.602	0.000
Effects from HARSH to MALAD				
Sum of indirect	1.009	0.262	3.855	0.000
Specific indirect				
MALAD				
REJECT				
HARSH	1.009	0.262	3.855	0.000

Standardized indirect effects

STANDARDIZED TOTAL, TOTAL INDIRECT, SPECIFIC INDIRECT, AND
DIRECT EFFECTS

STDYX Standardization


	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Effects from JUST to MALAD				
Sum of indirect	-0.190	0.039	-4.881	0.000
Specific indirect				
MALAD				
REJECT				
JUST	-0.190	0.039	-4.881	0.000
Effects from HARSH to MALAD				
Sum of indirect	0.147	0.037	4.012	0.000
Specific indirect				
MALAD				
REJECT				
HARSH	0.147	0.037	4.012	0.000

Cautionary note on testing indirect effects


All the tests for parameters and functions of parameters (e.g., indirect effects, R-square) are **Wald tests** and are thus based on the **assumption of normality**.

A mediation effect is a **product** of two (or more) parameters. Each parameter has a normal sampling distribution. Hence, the indirect effect is the product of two (or more) normally distributed variables.


$x \sim N(0,1), y \sim N(0,1)$




$x \sim N(2,1), y \sim N(0,1)$



$x \sim N(2,2), y \sim N(1,1)$



$x \sim N(2,2), y \sim N(-1,1)$



This is not normal!

As a result, the p-value you obtain with the Wald test is **incorrect** (may be too small or too large).

Solution: Bootstrapping

Bootstrapping is based on **resampling** from your data: For each bootstrap sample the model is estimated, including the **indirect effect**.

If we repeat this procedure many times (e.g., 1000 times), we obtain **1000 estimates** of the indirect effect, and this is used to determine the **95% confidence interval**.


Use:

```
ANALYSIS: BOOTSTRAP = 1000;
```

```
OUTPUT: Cinterval
```

See for details: Mplus User's Guide, p. 548-549 and Example 3.16.

(Note: bootstrapping requires the raw data.)



Multiple mediators

If you have **multiple mediators**, you may be interested in:

- **total** mediated effect
- **specific** mediated effect (through a particular mediator)

Mplus gives both.

When you have **multiple paths that include the same mediator**, you may be interested in all these paths simultaneously, or separately:

- **simultaneously**: y VIA mediator x
- **separately**: y IND mediator x


Take home message

- Crucial issue in multiple group CFA is **measurement invariance**
- Mplus imposes constraints for **strong factorial invariance**; note that this does not imply these constraints hold
- **You** have to test each constraint separately, using nested models and chi-square difference tests
- In multiple group analysis, the **mean structure** is of interest (in addition to the **covariance structure**)
- If a test/questionnaire is biased, you can **adjust** your model for this, and still **compare the groups in a fair & meaningful way**
- But when **too many items/subtests** are **biased**, it is hard/impossible to adjust for this.


Take home message

- **Path analysis** is another form of SEM
- It requires the use of the **structural equation**, which allows variables to be **regressed on each other** (using the **matrix B**)
- The measurement equation is only used to **transfer** the observed y variables to the structural equation (i.e., there is no “real” measurement model, since $\eta = y$)
- You may consider **different routes of mediation** in you model
- To test whether **indirect (i.e., mediated) effects** are significant, using **confidence intervals based on bootstrapping** should be preferred over the (standard) Wald test
- **Factor analysis (i.e., latent variables)** and **path analysis (i.e., structural relationships)** can also be combined in SEM

Cannabis use disorders



Body image and exercise



Emotional intelligence and life satisfaction

